


# 10 Datenbank in 20 Minuten erstellen

## Schritt 1: Überschriften / Spaltentitel im Tabellenblatt eintragen

	A	B	C	D	E	F	G
1	Titel	Autor	Gerne	Jahr	Auflage	Sprache(n)	Format
2							
3							

## Schritt 2: UserForm gestalten und Steuerelemente benennen




Steuerelement	Grundlagenvideo	Name	Caption
UserForm	-	UserForm	Literatur – Datenbank
TextBox	UserForm Folge 01	TextBox_Titel	-
TextBox	UserForm Folge 01	TextBox_Autor	-
ListBox	UserForm Folge 06	ListBox_Kategorie	-
ComboBox	UserForm Folge 02	ComboBox_Jahr	-
TextBox	UserForm Folge 01	TextBox_Auflage	-
SpinButton	UserForm Folge 05	SpinButton_Auflage	-
CheckBox	UserForm Folge 03	CheckBox_Deutsch	Deutsch
CheckBox	UserForm Folge 03	CheckBox_Englisch	Englisch
Frame	UserForm Folge 11	Frame_Format	Format
OptionButton	UserForm Folge 04	OptionButton1	Buch
OptionButton	UserForm Folge 04	OptionButton2	Ebook
OptionButton	UserForm Folge 04	OptionButton3	Hörbuch
CommandButton	UserForm Folge 01	Button_Eingabe	Eingabe
CommandButton	UserForm Folge 01	Button_Abbrechen	Abbrechen

## 10 Datenbank in 20 Minuten erstellen

## Schritt 3: Funktionen über VBA programmieren

- Ausgangswerte für die UserForm festlegen

```
Private Sub UserForm_Initialize ()  
  
    'Titel:  
    TextBox_Titel = "Buchtitel eingeben"  
    'Autor:  
    TextBox_Autor = "Autor eingeben"  
    'Kategorie:  
  
    With ListBox_Kategorie  
        .AddItem "Krimi"  
        .AddItem "Thriller"  
        .AddItem "Liebesroman"  
        .AddItem "Historischer Roman"  
        .AddItem "Sachbuch"  
        .AddItem "Fantasy"  
        .AddItem "..."  
    End With  
  
    'Jahr:  
    Dim i As Integer  
  
    With ComboBox_Jahr  
        For i = 1900 To 2017  
            .AddItem CStr(i)  
        Next  
    End With  
  
    'Auflage:  
    SpinButton_Auflage.Min = 1  
    SpinButton_Auflage.Value = 1  
    TextBox_Auflage.Text = SpinButton_Auflage.Value  
  
    'Sprache:  
    CheckBox_Deutsch.Value = False  
    CheckBox_Englisch.Value = False  
  
    'Format:  
    OptionButton1.Value = True  
  
End Sub
```





## 10 Datenbank in 20 Minuten erstellen

- Den Button "Eingabe" programmieren

```
Private Sub Button_Eingabe_Click()  
  
    'Erste freie Zeile ausfindig machen  
    Dim last As Integer  
  
    last = ActiveSheet.Cells(Rows.Count, 1).End(xlUp).Row + 1  
  
    'Informationen von der UserForm in das Tabellenblatt eintragen  
  
    'Titel:  
    Cells(last, 1).Value = TextBox_Titel  
    'Autor:  
    Cells(last, 2).Value = TextBox_Autor  
    'Kategorie:  
    Cells(last, 3).Value = ListBox_Kategorie.Value  
    'Jahr:  
    Cells(last, 4).Value = ComboBox_Jahr  
    'Auflage:  
    Cells(last, 5).Value = TextBox_Auflage.Value  
    'Sprache:  
    If CheckBox_Deutsch.Value = True Then Cells(last, 6).Value = _  
        CheckBox_Deutsch.Caption  
  
    If CheckBox_Englisch.Value = True Then Cells(last, 6).Value = _  
        Cells(last, 6).Value & " ," & CheckBox_Englisch.Caption  
  
    'Format:  
    If OptionButton1.Value = True Then Cells(last, 7).Value = "Buch"  
    If OptionButton2.Value = True Then Cells(last, 7).Value = "EBook"  
    If OptionButton3.Value = True Then Cells(last, 7).Value = "Hörbuch"  
  
End Sub
```

- Den SpinButton "Auflage" mit der TextBox verknüpfen

```
Private Sub SpinButton_Auflage_Change()  
  
    'Den SpinButton mit der TextBox verknüpfen  
    TextBox_Auflage.Text = SpinButton_Auflage.Value  
  
End Sub
```

- Den Button "Abbrechen" programmieren

```
Private Sub Button_Abbrechen_Click()  
  
    'UserForm schließen  
    Unload Me  
  
End Sub
```